Q:

How to display numbers as multiples of a square root

The extension field $\mathbb{Q}(\sqrt {2} + \sqrt {3})$ can be represented by a $4$ - dimensional vector space over $\mathbb{Q}$ with basis $\{1, \sqrt {2}, \sqrt {3}, \sqrt {6}\} $.
I made a matrix model of such a field where presenting the numbers is done as shown in the following example:
{1,2,3,4}.{1, Sqrt[2], Sqrt[3], Sqrt[6]}

$1 + 2\sqrt {2} + 3\sqrt {3} + 4\sqrt {6} $
The number {1,0,0,1/2} should be represented as:
$1 + \frac{1}{2}\sqrt {6} $
but is in fact represented as 
$1 + \sqrt{\frac{3}{2}}$
Is there a way to workaround this such that $ \{a,b,c,d \}$ is always represented as 
$a + b\sqrt {2} + c\sqrt {3} + d\sqrt {6} $
I suppose the title of the question should then be 'How to display numbers as multiples of a square root' ?

A:

Maybe you can use a wrapper + MakeBoxes. Here is the basis using the wrapper q:
QBasis = {1, q[Sqrt[2]], q[Sqrt[3]], q[Sqrt[6]]}

Now, we tell Mathematica how to format the q:
q /: MakeBoxes[q[x_], StandardForm] := MakeBoxes[x, StandardForm]
MakeBoxes[n_ q[x_], StandardForm] := RowBox[{
    Parenthesize[n, StandardForm, Times, Left],
    MakeBoxes[x]}
]

A couple examples:
{1, 0, 0, -1/2} . QBasis
{1, -I, 2.2, -2+I} . QBasis
{1, 0, 1, 1} . QBasis

1 - 1/2 Sqrt[6]
1 - I Sqrt[2] + 2.2 Sqrt[3] - (2 - I) Sqrt[6]
1 + Sqrt[3] + 1/2 Sqrt[6]

A:

A = {1, 2, 3, 1/2} /. {Rational[x_, y_] :>  If[x > 0, HoldForm[x/y], 
-HoldForm@Evaluate[-x/y]]};
B = MapAt[HoldForm, {1, Sqrt[2], Sqrt[3], Sqrt[6]}, 2 ;;]
A.B

1+2 Sqrt[2]+3 Sqrt[3]+1/2 Sqrt[6]

